
Semantic Web 1

1. Unit: SPARQL

Exercise 1.1 (SPARQL-Queries) Give SPARQL queries against mondial.n3 that yield an-
swers to the following questions:

• Name and population (ordered) of all countries that have more than 10.000.000 inhabitants.

• Name of all countries that have at least one city with more than 1.000.000 inhabitants.

• Name of all countries that have no city with more than 1.000.000 inhabitants.

• Name of all european countries that have no membership in the European Union.

• Abbreviations of all organizations whose headquarter is located in the capital of a member
country (together with the names of the country and the city).

# bigcountries.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT ?N ?P

FROM <file:mondial.n3>

WHERE {?X rdf:type mon:Country . ?X mon:name ?N . ?X mon:population ?P .

FILTER (?P > 10000000) }

ORDER BY DESC(?P)

# bigcities.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT DISTINCT ?N

FROM <file:mondial.n3>

WHERE {?X rdf:type mon:Country . ?X mon:name ?N .

?X mon:hasCity ?C . ?C mon:population ?P .

FILTER (?P > 1000000) }

# nobigcities.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT distinct ?N

FROM <file:mondial.n3>

WHERE {?X rdf:type mon:Country . ?X mon:name ?N .

OPTIONAL { ?X mon:hasCity ?C . ?C mon:population ?P . FILTER (?P > 1000000) } .

FILTER (!BOUND(?P)) }

# no-eu.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT distinct ?N

FROM <file:mondial.n3>

WHERE {?X rdf:type mon:Country . ?X mon:name ?N .

?X mon:encompassed [ mon:name ’Europe’ ] .

OPTIONAL { ?X mon:isMember [ mon:name ’European Union’] .

?X mon:name ?XX }

FILTER (!BOUND(?XX)) }

# bind XX (to anything ...) in case that EU membership is satisfied

# cap-hq.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

SELECT ?XN ?OA ?CN

FROM <file:mondial.n3>

WHERE {?O rdf:type mon:Organization . ?O mon:abbrev ?OA .

?O mon:hasHeadq ?C .

?X rdf:type mon:Country . ?X mon:carCode ?XN .

?X mon:capital ?C . ?C mon:name ?CN .

?X mon:isMember ?O }

ORDER BY ?CN

This is an example for a cyclic join: Organization - hasHeadq - City - isCapital - Country -
isMember - Organization. Note the occurrence of the join variable O that closes this circle. When
evaluated in the same order as stated in the query, the last triple pattern acts as a selection (the
actual evaluation order is defined by the optimizer).

Exercise 1.2 (SPARQL Formal Semantics) Consider the SPARQL Formal Semantics.

a) define a relational “null-tolerant join” that acts like ⊲⊳ above.

b) which SQL construct is similar to the “\” operator in the SPARQL algebra?

c) in the above algebra, OPT is expressed via left outer join, which is defined via “\” (while a
corresponding MINUS does not exist in the SPARQL syntax).
Such a MINUS (cf. part (b) of this exercise) provides a more intuitive idea of negation than “!
bound(x)”. Give a general pattern how to express (P1 MINUS P2) in SPARQL syntax.

d) recall the definition of ⊐⊲⊳ in the relational algebra (DB lecture) and define SPARQL’s ⊐⊲⊳ in
a similar way.

(parts of the solution are taken from [PAG06]: Jorge Pérez, Marcelo Arenas, Claudio Gutierrez:
Semantics and Complexity of SPARQL. International Semantic Web Conference 2006: 30-43, and
from [AG08]: Renzo Angles and Claudio Gutierrez: The Expressive Power of SPARQL. Interna-
tional Semantic Web Conference 2008: 114-129; use http://www.dblp.org)

a) Consider R(A,B,C) and S(A,B,D) where A is non-null, and B can contain nulls. Then, the
null-tolerant join ⊲⊳null can be defined by the following steps:

1) cartesian product of both relations, immediately evaluating the condition

r1.a = r2.a ∨ r1.a is null ∨ r2.a is null .

The result has the format [R1.A,R2.A,R1.B,R2.B, C,D].

– R1.A has always the same (non-null) value as R2.A.

– R1.B and R2.B can contain the same non-null-value, but also any of them can contain
a null value, while the other is also null, or contains a non-null value.

2) apply a projection that removes R2.A.

3) For handling B, a new basic operator has to be defined (similar to SQL’s binary “coalesce”
function: if the first argument is null, take the second one):

coalesce : ANY ×ANY, (v1, v2) 7→ v1 if v1 is not null,
(null, v) 7→ v

(note that coalesce(R1.B,R2.B) = coalesce(R2.B,R1.B) after evaluating the condition in
Step (1)).

b) SQL’s “WHERE NOT EXISTS ...” is similar.
Consider R1 and R2 as above.
SELECT * FROM R1 WHERE NOT EXISTS (

SELECT * FROM R2

WHERE R1.A = R2.A AND (R1.B = R2.B OR R1.B is null OR R2.B is null)).

c) (taken from [AG08], Section 3)
The basic idea is to replace (P1 MINUS P2) by

((P1 OPT P2) FILTER (!bound(?Y )))

where Y is a variable that occurs in P2, but not in P1.

Two more aspects have to be considered:

• If P2 is of the form (P ′

2
OPT P ′′

2
), then Y must be a variable from P ′

2
– i.e., a non-optional

variable (otherwise there are solutions to P2 that do not bind it).

• If there is no such variable (i.e. all non-optional variables of P2 occur also in P1, one must
introduce one: take any non-optional triple pattern T that

i) contains at least one new variable X ′ and

ii) is sure to be satisfied whenever (P1 and) P2 is satisfied (i.e., it can be a renamed copy
(?X q ?X ′) of some triple pattern (?X q ?Z) from P1, or any arbitrary pattern that is
known to be satisfied in the application)

and use ((P1 OPT (P2 AND T )) FILTER (!bound(X ′))) .

In practice, instead of (X ′ p t), any pattern can be used where it is sure that it is also bound
for each match of P2.

Example: all cities in a country which are not its capital:

# cities-not-capital.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

select ?X ?C

from <file:mondial.n3>

where { ?X a mon:Country; mon:hasCity ?C

OPTIONAL {?X mon:capital ?C . ?X mon:capital ?C2}

FILTER (! bound(?C2)) }

X and C occur in P1 and in P2, so a (useless) triple pattern is added to bind C2.

d) Ω1 ⊐⊲⊳Ω2 = (Ω1 ⊲⊳ Ω2)∪(Ω1\s(Ω1 ✄< Ω2)) where the semijoin is defined as usual as Ω1 ✄< Ω2 =
π[var(Ω1)](Ω1 ⊲⊳ Ω2), and \s denotes the classical set difference from set algebra.

Note that it is not necessary to extend the second part of the union with null values, as it is
done in the relational algebra.

Exercise 1.3 (Outer Join) Recall that SPARQL’s OPTIONAL corresponds to a left outer join.

• Give a general pattern how to express a full outer join (i.e., “outer” to both sides) in the
SPARQL algebra (consider as input two mappings R and S and give an expression for R⊐⊲⊳⊏S)
and in SPARQL.

• Give all cities (name as ?XN) that are the capital of a country (:capital) or that are located at
a river (:locatedAt) or both (return the names ?CN of the country and/or the river (?RN)).

• Replace the full outer join by a two left outer joins: (R⊐⊲⊳S)∪(S ⊐⊲⊳R). Note that the intersec-
tion of both subterms is the inner join. With set semantics, these duplicates are automatically
removed. Otherwise apply a DISTINCT.

Alternatively, the inner join can be removed from the second term:

(R⊐⊲⊳S) ∪ ((S ⊐⊲⊳R) \s (S ⊲⊳ R))
or (R⊐⊲⊳S) ∪ ((S ⊐⊲⊳R) \ (S ✄< R))

(recall that \ denotes the not-exists-like operator from the SPARQL algebra, and \s denotes
the classical set difference).

For SPARQL, the query is of the form

DISTINCT ... WHERE { { P_R(X) OPTIONAL P_S(Y) }

UNION

{ P_S(Y) OPTIONAL P_R(X) }

or

... WHERE { { P_R(X) OPTIONAL P_S(Y) }

UNION

{ P_S(Y) OPTIONAL P_R(X) FILTER !bound(X) } }

• There is an intuitive solution that replaces the outer join by two optionals: take a city, and if
it is a capital, list the country, and if it is located at a river, list the river:

# capitals-at-rivers-1.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

select ?XN ?CN ?RN

from <file:mondial.n3>

where { ?X a mon:City ; mon:name ?XN.

OPTIONAL { ?C a mon:Country; mon:name ?CN; mon:capital ?X }

OPTIONAL { ?X mon:locatedAt ?R . ?R a mon:River; mon:name ?RN} }

The query yields one line for each city, including those that are neither capitals, nor located at
a river.

The second solution applies the solution of (a):

# capitals-at-rivers-2.sparql

prefix mon: <http://www.semwebtech.org/mondial/10/meta#>

prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

select ?XN ?CN ?RN

from <file:mondial.n3>

where {{ ?X a mon:City ; mon:name ?XN .

?C a mon:Country; mon:name ?CN; mon:capital ?X

OPTIONAL { ?X mon:locatedAt ?R . ?R a mon:River; mon:name ?RN } }

UNION

{ ?X a mon:City ; mon:name ?XN .

?X mon:locatedAt ?R . ?R a mon:River; mon:name ?RN

OPTIONAL { ?C a mon:Country; mon:name ?CN; mon:capital ?X }

FILTER (!bound(?C)) }}

Exercise 1.4 (SPARQL Formal Semantics: OPTIONAL) Consider the SPARQL Formal
Semantics.

Prove or show a counterexample:

The statement

If OPT(A, B) is an optional graph pattern, where A and B are graph patterns, then S is a
solution of OPT(A,B) if S is a pattern solution of A and of B otherwise if S is a solution to
A, but not to A and B.

describes the same semantics as above.

The given characterization is the one from the W3C SPARQL Recommendation from 20061004.
The counterexample is taken from [PAG06], Examples 1 and 3:

RDF database D:

D = { (B1 name paul), (B1 phone 777-3426),
(B2 name john), (B2 email john@acd.edu),
(B3 name george), (B3 webPage www.george.edu),
(B4 name ringo), (B4 email ringo@acd.edu),
(B4 email www.starr.edu), (B4 phone 888-4537) }

Query pattern:

P = ((?X ,name, paul) OPT ((?Y , name, george) OPT (?X , email, ?Z))) =: (P1 OPT (P2 OPT P3)) .

[[P1]] = {{X/B1}}.
[[P2]] = {{Y/B3}}.
[[P3]] = {{X/B2, Z/john@}, {X/B4, Z/ringo@}}.
[[P2 OPT P3]] = [[P2 ⊐⊲⊳P3]] = {{X/B2, Y/B3, Z/john@}, {X/B4, Y/B3, Z/ringo@}}.
[[P ]] = [[P1]]⊐⊲⊳[[P2 ⊐⊲⊳P3]] = {{X/B1}}.

On the other hand according to the textual W3C characterization, S := {{X/B1, Y/B3}} is a
solution to P : S := {{X/B1, Y/B3}} is a solution to P1 and to P2 OPT P3; the latter holds since
it is a solution to P2, although not to P3.

The counterexample exploits the fact that it is not well-designed (i.e., X occurs inside the inner
optional, and in the outermost pattern, but not directly outside the inner optional).

Note that the “declarative”, but non-algebraic W3C characterization is also problematic from the
operational aspect since the solution must first be guessed before being tested. An algebraic (and
thus compositional) semantics allows a bottom-up computation from inside-out.

Exercise 1.5 (SPARQL: Filter-Safe Expressions) Consider the following definition:

Definition 1 ([PAG06, AG 08 )]A SPARQL expression is filter-safe, if for every subexpression
of the form (P FILTER R), var(R) ⊆ var(P ).

Filters of the forbidden from are rather commonly used, e.g.,

{ ?P1 a :Person; :age ?A1. ?P2 a :Person

OPTIONAL { ?P2 :age ?A2 . FILTER ( ?A2 > ?A1 ) }}

a) Sketch an algorithm that rewrites non-filter safe queries into safe ones. First, try it on your
own, then maybe look in [AG08].

b) Is there a similar thing in SQL and the relational algebra?

a) An intuitive solution is to duplicate (relevant parts of) the pattern outside the OPTIONAL for
binding of the variable to inside:

{ ?P1 a :Person; :age ?A1. ?P2 a :Person

OPTIONAL { ?P1 a :Person; :age ?A1 . ?P2 a :Person . ?P2 :age ?A2 . FILTER ( ?A2 > ?A1 ) } }

The variables occurring in the inner pattern and in the outer pattern act as join variables when
computing the outer join.

The complete algorithm can be found in [AG08]. There, the fourth if-case (lines 7-10) cover
the above case. The fifth if-case (lines 12-24) just throws an error in case a filter is still unsafe.

Note (DB Theory lecture): consider the similarity with the transformation of general formulas
into RANF when moving conjuncts into negated subformulas to make them self-contained.

b) Yes: correlated subqueries. For instance: “all countries where more than 25% of the population
live in the capital”:

SELECT name FROM country

WHERE population < 4 *

(SELECT population FROM city

WHERE city.name= country.capital AND country.province=city.province

AND city.country= country.code)

Here, the corresponding algebra expression also “copies” the country columns into the subquery
by using a broad join.


